R. Brigola, TH Nürnberg Georg Simon Ohm, 2019

Mathematica - Notebooks als Bonusmaterial zum Lehrbuch

[1] Rolf Brigola Fourier-Analysis und Distributionen,
Eine Einführung mit Anwendungen,
edition swk, Hamburg 2019

Demonstrationsbeispiele zu Darstellungen trigonometrischer Polynome mit *Mathematica*

1. Darstellung eines komplexwertigen trigonometrischen Polynoms als zirkuläre Schwingung

Als Beispiel sei betrachtet

$$P(t) = I/2\ \mathrm{Sin}[\omega_0 t] + I\ \mathrm{Sin}[2\ \omega_0 t] - \mathrm{Cos}[3\ \omega_0 t]$$

für $\omega_0 = \pi/2$, Periode $T = 4$

In[1]:=
```
ω0 = Pi / 2;  T = 2 Pi / ω0;
P[t_] = I / 2 Sin[ω0 t] + I Sin[2 ω0 t] - Cos[3 ω0 t]
Table[{t, Re[P[t]], Im[P[t]]}, {t, 0, 6, 0.015}] //
 Graphics3D[{{Red, Lighting → "Neutral", Tube@#},
    {RGBColor[0, 1, 5], Arrow@Tube[{{0, 0, 0}, 1.5 #}, 0.02] & /@
      {{4, 0, 0}, {0, 1, 0}, {0, -1, 0}, {0, 0, 1}, {0, 0, -1}}},
    Arrowheads[1 / 40, Appearance → "Projected"], AbsoluteThickness[3],
    Opacity[0.5], Arrow[{# {1, 0, 0}, #}] & /@ #[[ ;;  ;; 10]]}, Boxed → False,
   Axes → True, AxesLabel → {"t", "Re(P(t))", "Im (P(t))"}, LabelStyle →
    Directive[Black, FontFamily → "Times", FontSize -> 18], ImageSize → Large] &
```

Out[2]= $-\mathrm{Cos}\left[\frac{3\pi t}{2}\right]+\frac{1}{2}\, i\, \mathrm{Sin}\left[\frac{\pi t}{2}\right]+i\,\mathrm{Sin}[\pi t]$

Out[3]=



2. Darstellung des gleichen Beispiels als Ortskurve in der komplexen Ebene über eine Periode: Die Ortskurve beginnt bei (-1,0) und wird mit wachsendem t farblich von rot über blau, grün nach schwarz durchlaufen. Farbwechsel jeweils nach Dauer T/4.

In[ ]:=
```
Table[{P[T / 4], P[T / 2], P[3 T / 4], P[T]}]
```

Out[ ]= $\left\{\frac{i}{2}, 1, -\frac{i}{2}, -1\right\}$

In[ ]:=
```
plot1 = ParametricPlot[{Re[P[t]], Im[P[t]]}, {t, 0, T / 2},
    ColorFunction → Function[{x, y, t}, If[t < T / 4, Red, Blue]],
    ColorFunctionScaling → False, PlotRange → All];
plot2 = ParametricPlot[{Re[P[t]], Im[P[t]]}, {t, T / 2, T},
    ColorFunction → Function[{x, y, t}, If[t < 3 T / 4, Green, Black]],
    ColorFunctionScaling → False, PlotRange → All];
Show[{plot1, plot2}]
```

Out[ ]=


3. Darstellung von Realteil (blau) und Imaginärteil (rot) getrennt über eine Periodendauer

In[ ]:=
```
plot3 = Plot[Re[P[t]], {t, 0, T}, PlotStyle -> Directive[Blue], PlotRange → All];
```

In[ ]:=
```
plot4 = Plot[Im[P[t]], {t, 0, T}, PlotStyle -> Directive[Red], PlotRange → All];
```

In[ ]:=
```
Show[{plot3, plot4}]
```

Out[ ]=


4. Darstellung von Betragsverlauf (blau) und Phasenverlauf (rot) über eine Periodendauer bei Verwendung des Hauptwerts für das Argument, d.h. mit Werten zwischen - π und π

In[ ]:=
```
plot5 = Plot[Abs[P[t]], {t, 0, T}, PlotStyle -> Directive[Blue], PlotRange → {0, 2}]
```

Out[ ]=


In[ ]:= `plot6 = Plot[Arg[P[t]], {t, 0, T}, PlotStyle -> Directive[Red], PlotRange → {-6 / 5 Pi, 6 / 5 Pi}]`

Out[ ]=



Um Ihre Vorstellungskraft zu stärken, versuchen Sie, einzelne Werte und Verläufe der Funktion (Realteil, Imaginärteil, Betrag, Phase), die Sie in einer der Darstellungen auswählen, mit ihren Entsprechungen in den anderen Darstellungen zu identifizieren und zu überprüfen.

Nur als einfaches Beispiel etwa:

In[ ]:=
```
P[T / 4]
Re[P[T / 4]]
Im[P[T / 4]]
Abs[P[T / 4]]
Arg[P[T / 4]]
```

Out[ ]= $\frac{i}{2}$

Out[ ]= 0

Out[ ]= $\frac{1}{2}$

Out[ ]= $\frac{1}{2}$

Out[ ]= $\frac{\pi}{2}$